Black-Scholes Call Derivation via Expectation 01-12-11.nb

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing "shift" + "enter". Respond "Yes" in response to the query to evaluate initialization cells.

In[125]:=

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

The formula for a European call option is derived by calculating the expected value of the payoff under the risk-neutral probability measure. This approach to option valuation goes by various names, including the martingale method. It is also demonstrated that the formula for a put option can be obtained from put-call parity.

I refine work in *Book Black-Scholes Call Derivation 06-08-10* and *Option Derivation 01-14-08* and similar calculations from much earlier.

Under the martingale approach to option valuation (e.g. *Martingale Methods in Financial Modelling, Musiela and Rutkowski*), the fair value of a European call option involves the calculation of an expectation

$$C(K,\ t) = \mathbb{E}_Q\left[e^{-r(T-t)}\, C(S_T,\ K,\ T)\right]$$

where the expectation is taken under the risk neutral probability measure Q. For constant risk free interest rate,

$$C(K,\ t) = e^{-r(T-t)}\, \mathbb{E}_Q[\max(S_T - K,\ 0)]$$

Note

2/29/2016 8:57 PM Since the calculation below was originally performed, it has become a "one-liner" as a consequence of improvements to *Mathematica*. See the Black-Scholes example under Expectation/Ap·

plications/Finance.

Nonetheless, it is worthwhile to see some intermediate steps. For payoffs more complicated that a simple call option, a step by step procedure will likely be required.

Some notational preliminaries

I invoke the *Notation* package so that details of the calculation more closely resemble the form they would take in the financial literature.

In[127]:=
```
<< Notation`
```

In[128]:=
```
(* suppress some annoying warning messages *)Off[Solve::"ifun"];
Off[Symbolize::"bsymbexs"];
```

In[130]:=
```
Symbolize[ 𝔼Q ];
Symbolize[ fQ ];
Symbolize[ Ct ];
Symbolize[ Pt ];
Symbolize[ ST ];
Symbolize[ St ];
Symbolize[ ϵK ];
```

Derivation of fair value for a European style exercise call option

Here, I focus solely on the details of the calculation of the formula for the fair value. A reader unfamiliar with option valuation theory should definitely read background material such as the aforementioned text by Musiella and Rutkowski. The calculation of the fair value of an option is only the tip of an iceberg of deeper knowledge and insights to be gained from understanding the theory.

Define the expectation to be calculated

In[137]:=
```
w[1] = Ct == Exp[-r (T - t)] 𝔼Q[Max[ST[ϵ] - K, 0] ]
```

Out[137]=
$$C_t = e^{-r(-t+T)}\,\mathbb{E}_Q[\text{Max}[0, -K + S_T[\epsilon]]]$$

Assume that the stock dynamics are described by geometric Brownian motion

In[138]:=
$$w[2] = S_T[\epsilon] == S_t \,\text{Exp}\left[\left(\mu - \frac{\sigma^2}{2}\right)(T-t) + \epsilon\,\sigma\sqrt{(T-t)}\right]$$

Out[138]=
$$S_T[\epsilon] == e^{\sqrt{-t+T}\,\epsilon\,\sigma + (-t+T)\left(\mu - \frac{\sigma^2}{2}\right)} S_t$$

where ϵ is a Weiner process. Introduce some simplifying notation

In[139]:=
$$w[3] = w[1] \;/.\; (w[2] \;/.\; \text{Equal} \to \text{Rule}) \;/.\; T \to \tau + t \;/.\; \mu \to r - q$$

Out[139]=
$$C_t == e^{-r\,\tau}\,\mathbb{E}_Q\left[\text{Max}\left[0, -K + e^{\epsilon\,\sigma\sqrt{\tau} + \left(-q + r - \frac{\sigma^2}{2}\right)\tau} S_t\right]\right]$$

I use Int instead of Integrate to temporarily suppress the evaluation of the integral.

In[140]:=
$$w[4] = w[3] \;/.\; \mathbb{E}_Q[a\_] \to \text{Int}[f_Q[\epsilon]\, a, \{\epsilon, -\infty, \infty\}]$$

Out[140]=
$$C_t == e^{-r\,\tau}\,\text{Int}\left[f_Q[\epsilon]\,\text{Max}\left[0, -K + e^{\epsilon\,\sigma\sqrt{\tau} + \left(-q + r - \frac{\sigma^2}{2}\right)\tau} S_t\right], \{\epsilon, -\infty, \infty\}\right]$$

I have used my mouse to “cut” the integrand of w[4] and then paste it into the next expression. This task could have accomplished programmatically with w[4]⟦2, 2, 1, 2, 2⟧

In[141]:=
$$w[5] = -K + e^{\epsilon\,\sigma\sqrt{\tau} + \left(-q + r - \frac{\sigma^2}{2}\right)\tau} S_t$$

Out[141]=
$$-K + e^{\epsilon\,\sigma\sqrt{\tau} + \left(-q + r - \frac{\sigma^2}{2}\right)\tau} S_t$$

For what value of ϵ is this zero?

In[142]:=
$$w[6] = \text{Solve}[w[5] == 0, \epsilon][[1, 1]] \;/.\; \epsilon \to \epsilon_K$$

Out[142]=
$$\epsilon_K \to \text{ConditionalExpression}\left[\left(2\,q\,\tau - 2\,r\,\tau + \sigma^2\,\tau + 2\left(2\,i\,\pi\,C[1] + \text{Log}\left[\frac{K}{S_t}\right]\right)\right) \Big/ \left(2\,\sigma\sqrt{\tau}\right), C[1] \in \text{Integers}\right]$$

Choose the principal branch

In[144]:=
$$w[6] = w[6] \;/.\; C[1] \to 0$$

Out[144]=
$$\epsilon_K \to \left(2\,q\,\tau - 2\,r\,\tau + \sigma^2\,\tau + 2\,\text{Log}\left[\frac{K}{S_t}\right]\right) \Big/ \left(2\,\sigma\sqrt{\tau}\right)$$

With knowledge of range of ϵ over which the integrand is non zero, I manipulate the integral expression

In[145]:=
$$w[7] = w[4] \;/.\; \text{Max}[0, a\_] \to a \;/.\; \{\epsilon, -\infty, \infty\} \to \{\epsilon, \epsilon_K, \infty\}$$

Out[145]=
$$C_t == e^{-r\,\tau}\,\text{Int}\left[\left(-K + e^{\epsilon\,\sigma\sqrt{\tau} + \left(-q + r - \frac{\sigma^2}{2}\right)\tau} S_t\right) f_Q[\epsilon], \{\epsilon, \epsilon_K, \infty\}\right]$$

and introduce the explicit distribution (normal distribution for a Weiner process)

In[146]:=
```
w[8] = w[7] /. f_Q[ϵ] → 1/Sqrt[2 π] Exp[-ϵ^2/2]
```

Out[146]=

$$C_t == e^{-r\tau}\,\mathrm{Int}\left[\frac{1}{\sqrt{2\pi}} e^{-\frac{\epsilon^2}{2}}\left(-K + e^{\epsilon\,\sigma\sqrt{\tau}+\left(-q+r-\frac{\sigma^2}{2}\right)\tau} S_t\right),\ \{\epsilon,\ \epsilon_K,\ \infty\}\right]$$

It’s time to let *Mathematica* perform the integral.

In[147]:=
```
w[9] = w[8] /. Int → Integrate // ExpandAll
```

Out[147]=

$$C_t == -\frac{1}{2} e^{-r\tau} K\,\mathrm{Erfc}\left[\frac{\epsilon_K}{\sqrt{2}}\right] + \frac{1}{2} e^{-q\tau} S_t\,\mathrm{Erfc}\left[\frac{\epsilon_K}{\sqrt{2}} - \frac{\sigma\sqrt{\tau}}{\sqrt{2}}\right]$$

The cumulative standard normal distribution is preferred over erf in the financial literature.

In[148]:=
```
w[10] = w[9] //. {Erfc[x_] → 1 - Erf[x], Erf[x_] → 2 𝒩[Sqrt[2] x] - 1} // ExpandAll
```

Out[148]=

$$C_t == -e^{-r\tau} K + e^{-q\tau} S_t + e^{-r\tau} K\,\mathcal{N}[\epsilon_K] - e^{-q\tau} S_t\,\mathcal{N}\left[\epsilon_K - \sigma\sqrt{\tau}\right]$$

To obtain the familiar form, more manipulation is required.

In[149]:=
```
w[11] = w[10] /. 𝒩[x_] → 1 - 𝒩[-x] // Expand
```

Out[149]=

$$C_t == -e^{-r\tau} K\,\mathcal{N}[-\epsilon_K] + e^{-q\tau} S_t\,\mathcal{N}\left[-\epsilon_K + \sigma\sqrt{\tau}\right]$$

with

In[150]:=
```
w[6]
```

Out[150]=

$$\epsilon_K \to \left(2\,q\,\tau - 2\,r\,\tau + \sigma^2\,\tau + 2\,\mathrm{Log}\left[\frac{K}{S_t}\right]\right) \Big/ \left(2\,\sigma\sqrt{\tau}\right)$$

which is the classical Black-Scholes formula for the fair value of a call option. This derivation should be compared to the calculation involving direct solution of the Black-Scholes partial differential equation with boundary and terminal conditions corresponding to the pay of a call option, which will be presented in a different notebook.

Derivation of formula for put option

An analogous calculation could be performed to obtain the formula for the fair value of a put option. However, that result can be obtained more quickly by considerations of symmetry via a relationship called (put/call parity)

Consider the following identity

In[151]:=
```
wp[1] = Max[S_T - K, 0] + Max[K - S_T, 0] == S_T - K
```
Out[151]=

$$\text{Max}[0, K - S_T] + \text{Max}[0, -K + S_T] == -K + S_T$$

For convenience, convert this equation to an expression

In[152]:=
```
wp[2] = wp[1] /. Equal → Subtract
```
Out[152]=

$$K - S_T + \text{Max}[0, K - S_T] + \text{Max}[0, -K + S_T]$$

Apply the risk neutral expectation operator

In[153]:=
```
wp[3] = (𝔼_Q[Exp[-r τ] #]) & /@ wp[2] // ExpandAll
```
Out[153]=

$$\mathbb{E}_Q\left[e^{-r\tau} K\right] + \mathbb{E}_Q\left[-e^{-r\tau} S_T\right] + \mathbb{E}_Q\left[e^{-r\tau}\,\text{Max}[0, K - S_T]\right] + \mathbb{E}_Q\left[e^{-r\tau}\,\text{Max}[0, -K + S_T]\right]$$

Identify the put and call terms

In[154]:=
```
wp[4] = wp[3] /. 𝔼_Q[ⅇ^(-r τ) Max[0, K - S_T]] → P_t /. 𝔼_Q[ⅇ^(-r τ) Max[0, -K + S_T]] → C_t
```
Out[154]=

$$C_t + P_t + \mathbb{E}_Q\left[e^{-r\tau} K\right] + \mathbb{E}_Q\left[-e^{-r\tau} S_T\right]$$

The remaining terms are not stochastic

In[155]:=
```
wp[5] = wp[4] /. 𝔼_Q[x_] /; FreeQ[x, ϵ] → x
```
Out[155]=

$$C_t + e^{-r\tau} K + P_t - e^{-r\tau} S_T$$

Solve for the put

In[156]:=
```
wp[6] = Solve[wp[5] == 0, P_t][[1, 1]] /. Rule → Equal
```
Out[156]=

$$P_t == -e^{-r\tau}\left(C_t\, e^{r\tau} + K - S_T\right)$$

Thus, the fair value of a put option is conveniently expressed in terms of previous derived result for the call option.